Option Explicit

Sub Créat_Nomenclature()
    On Error GoTo ErrorHandler
    
    Dim swApp As SldWorks.SldWorks
    Dim swModel As SldWorks.ModelDoc2
    Dim swDraw As SldWorks.DrawingDoc
    Dim swActiveView As SldWorks.View
    Dim swBOMTable As SldWorks.BomTableAnnotation
    Dim swBOMFeat As SldWorks.BomFeature
    Dim swFeat As SldWorks.Feature
    Dim Config As String
    Dim TemplateName As String
    Dim myModelView As Object
    Dim Part As Object
    Dim myVar As Variant
    Dim myRev As String
    Dim chemin As String
    Dim nomFichier As String
    
    ' Initialisation
    Set swApp = Application.SldWorks
    Set swModel = swApp.ActiveDoc
    
    ' Vérification qu'un document est ouvert
    If swModel Is Nothing Then
        MsgBox "Aucun document actif ouvert.", vbExclamation
        Exit Sub
    End If
    
    ' Vérification que c'est un plan
    If swModel.GetType <> swDocDRAWING Then
        MsgBox "Le document actif doit être un plan.", vbExclamation
        Exit Sub
    End If
    
    Set swDraw = swModel
    Set swActiveView = swDraw.GetFirstView
    Set swActiveView = swActiveView.GetNextView
    
    If swActiveView Is Nothing Then
        MsgBox "Aucune vue trouvée dans le plan.", vbExclamation
        Exit Sub
    End If
    
    Config = swActiveView.ReferencedConfiguration
    Set Part = swApp.ActiveDoc
    Set myModelView = Part.ActiveView
    
    ' Maximiser la fenêtre
    myModelView.FrameState = swWindowState_e.swWindowMaximized
    
    ' Récupérer la révision
    myRev = Part.GetCustomInfoValue("", "Révision")
    If myRev = "" Then myRev = "A" ' Valeur par défaut
    
    ' Chemins
    TemplateName = "E:\Modele SW\nomenclature_Sylob2.sldbomtbt"
    chemin = "\\SRV-DATA\Data\COMMUN\Y- DIFFUSION PLANS\0-Plan en Attente\0- Fichiers pour extranet à l'indice actuel"
    
    ' Vérifier que le modèle existe
    If Dir(TemplateName) = "" Then
        MsgBox "Le modèle de nomenclature n'existe pas :" & vbCrLf & TemplateName, vbExclamation
        Exit Sub
    End If
    
    ' Extraire le nom du fichier
    myVar = Split(Part.GetPathName, "\", -1)
    If UBound(myVar) < 0 Then
        'MsgBox "Impossible de déterminer le nom du fichier.", vbExclamation
        Exit Sub
    End If
    
    nomFichier = Mid(myVar(UBound(myVar)), 1, Len(myVar(UBound(myVar))) - 7) & "-" & myRev & "-nom.csv"
    
    ' Création de la nomenclature
    Set swBOMTable = swActiveView.InsertBomTable2(False, 0, 0.2, _
                                                   swBOMConfigurationAnchorType_e.swBOMConfigurationAnchor_TopLeft, _
                                                   swBomType_e.swBomType_TopLevelOnly, _
                                                   Config, TemplateName)
    
    If swBOMTable Is Nothing Then
        MsgBox "Erreur lors de la création de la nomenclature.", vbCritical
        Exit Sub
    End If
    
    
    
    
    ' Obtenir la feature BOM
    Set swBOMFeat = swBOMTable.BomFeature
    If swBOMFeat Is Nothing Then
        MsgBox "Impossible d’accéder à la feature BOM.", vbCritical
        Exit Sub
    End If

    ' Activer "Suivre l’ordre de l’assemblage"
    On Error Resume Next
    swBOMFeat.FollowAssemblyOrder2 = True
    If Err.Number <> 0 Then
        MsgBox "Erreur : la propriété FollowAssemblyOrder2 n’est pas accessible.", vbCritical
        Err.Clear
    End If
    On Error GoTo 0

    ' Rafraîchir
    swModel.ForceRebuild3 False

    MsgBox "Nomenclature créée avec l’option 'Suivre l’ordre de l’assemblage' activée.", vbInformation
    
    
     
    
    
    
    
    
    
    ' Export vers CSV
    WriteBom chemin & "\" & nomFichier, swBOMTable
    
    ' Suppression de la nomenclature
    Set swBOMFeat = swBOMTable.BomFeature
    Set swFeat = swBOMFeat.GetFeature
    swFeat.Select2 False, -1
    swModel.EditDelete
    
   ' MsgBox "Nomenclature exportée avec succès :" & vbCrLf & nomFichier, vbInformation
    Exit Sub
    
ErrorHandler:
    MsgBox "Erreur " & Err.Number & " : " & Err.Description, vbCritical
End Sub

Sub WriteBom(FilePath As String, Table As SldWorks.TableAnnotation)
    On Error GoTo ErrorHandler
    
    Dim N As Integer
    Dim i As Integer
    Dim j As Integer
    Dim myArray() As String
    Dim ligne As String
    
    N = FreeFile
    Open FilePath For Output As #N
    
    ReDim myArray(Table.ColumnCount - 1)
    
    For i = 0 To Table.RowCount - 1
        For j = 0 To Table.ColumnCount - 1
            ' Nettoyer le texte (enlever retours à la ligne, guillemets)
            myArray(j) = Replace(Table.Text(i, j), vbCrLf, " ")
            myArray(j) = Replace(myArray(j), vbCr, " ")
            myArray(j) = Replace(myArray(j), vbLf, " ")
            myArray(j) = Replace(myArray(j), """", """""") ' Échapper les guillemets
            
            ' Ajouter des guillemets si nécessaire
            If InStr(myArray(j), ";") > 0 Or InStr(myArray(j), ",") > 0 Then
                myArray(j) = """" & myArray(j) & """"
            End If
        Next
        
        ' Utiliser point-virgule pour CSV (standard européen)
        Print #N, Join(myArray, ";")
    Next
    
    Close #N
    Exit Sub
    
ErrorHandler:
    If N > 0 Then Close #N
    MsgBox "Erreur lors de l'écriture du fichier CSV : " & Err.Description, vbCritical
End Sub



